


Handleiding : Snake



In deze handleiding wordt stap voor stap uitgelegd hoe je het spelletje Snake, in Scratch maakt.



Uitleg over handleiding

Na elke stap kun je de code testen door op het groene vlaggetje te klikken.

Rode blokken voor de code geven aan waar nieuwe code is toegevoegd. Rode kringen over de of code geven aan waar de code is aangepast. Een stuk code zonder rode indicaties is helemaal nieuw.


Mocht je de omschrijving niet helemaal begrijpen, de afbeeldingen van de code laten zien hoe je code er aan het einde van elke stap uit ziet.

Waar in handleiding verwezen wordt naar een bestandslocatie is dit: _____

Website Scratch: <https://scratch.mit.edu/>

Voor Scratch 2 is er ook een offline editor :

<https://scratch.mit.edu/download>




1

Het laden van de Sprites

Voordat we code kunnen gaan schrijven moeten we eerst de juiste sprites laden.

1. Klik met rechtermuis toets op de kat en kies verwijderen
2. Klik op laad sprite en laad afbeelding slang-01 ### pad toevoegen
3. Klik op Uiterlijken
4. Klik op Upload uiterlijk om afbeelding slang-02 te laden ### pad toevoegen
5. Klik op het poppetje om de de Scratch Sprite bibliotheek te openen, en kies de sinaasappel (orange) door er dubbel op te klikken.



2


Een vast begin

Bij elke start moet het spel op dezelfde manier beginnen. De slang start op dezelfde positie, met hetzelfde uiterlijk en is elke keer even groot.

Sleep vanuit **Gebeurtenissen** het **wanneer wordt aangeklikt** blok naar het code veld.

Voeg daar vanuit **Uiterlijken** de blokken **maak grootte** en **verander uiterlijk** naar aan toe. En vul de waarden in zoals ze in de afbeelding hiernaast staan.

Voeg daarna vanuit **Beweging** de blokken **ga naar** en **richt naar** toe. Vul hier overal de waarde **0** in.



3

Beweging

De slang moet natuurlijk gaan bewegen. Met een constante snelheid vooruit gaan. Daarvoor moeten we de slang snel achter elkaar telkens een klein stukje verplaatsen. Om dit te doen gebruiken we een **herhaal**. Ons programma blijft de code in een herhaal lus herhalen, totdat het programma gestopt wordt.

In de herhaal lus verplaatsen we de sprite een klein stukje met een **neem 10 stappen** blok. Om te voorkomen dat het spel veels te moeilijk wordt voegen we er een **wacht** blok, met als waarde 0.1, aan toe.



4

Besturing

We willen de slang besturen met de pijltjes toetsen. In de bewegings herhaal staat een wacht. Als de herhaal toevallig net in de wacht staat als wij een toets indrukken mist Scratch onze invoer. Maar gelukkig kunnen in Scratch verschillende groene vlaggen naast elkaar staan.

We maken dus een nieuw code blok, een **Wanneer groene vlag**, met een **herhaal** en in de herhaal een **besturen → Als dan** blok. Het als dan blok heeft een voorwaarde nodig. Wij voegen een

Waarnemen → Als toets ingedrukt

in. Als we op het zwarte driehoekje klikken, krijgen we een lijst met toetsen. Als toets kiezen wij pijltje links.

In de als dan plaatsen we nu een **beweging → richt naar** blok en maken de waarde -90 (links).



5

Alle kanten op

Nu kunnen we dus naar links. Maar we willen ook naar rechts, boven en onder kunnen. Klik met de rechtermuis knop op het als dan en kies kopie maken. Voeg de kopie toe onder de eerste als dan. Dit doen we nog 2 keer. Pas de waarden voor toets en richting vervolgens aan zodat we vier kanten op kunnen.

```
wanneer vlag wordt aangeklikt
herhaal
  als toets pijltje links ingedrukt? dan
    richt naar -90 graden
  als toets pijltje rechts ingedrukt? dan
    richt naar 90 graden
  als toets pijltje omhoog ingedrukt? dan
    richt naar 0 graden
  als toets pijltje omlaag ingedrukt? dan
    richt naar 180 graden
```

6

Slang code tot nu toe

Klopt de code die je hebt geschreven met de code hieronder?

```
wanneer vlag wordt aangeklikt
  maak grootte 20 %
  verander uiterlijk naar slang-01
  ga naar x: 0 y: 0
  richt naar 0 graden
  herhaal
    neem 10 stappen
    wacht 0.1 sec.
```

```
wanneer vlag wordt aangeklikt
herhaal
  als toets pijltje links ingedrukt? dan
    richt naar -90 graden
  als toets pijltje rechts ingedrukt? dan
    richt naar 90 graden
  als toets pijltje omhoog ingedrukt? dan
    richt naar 0 graden
  als toets pijltje omlaag ingedrukt? dan
    richt naar 180 graden
```



7

De slang

Leuk zo'n bolletje dat we kunnen besturen maar we zouden een slang maken.

Om van het bolletje een slang te maken we, vlak voordat we het bolletje verplaatsen, een kopie van het bolletje, die we laten staan.

Dit doen we met behulp van het commando **besturen** → **maak een kloon**.




```
wanneer vlag wordt aangeklikt
  maak grootte 20 %
  verander uiterlijk naar slang-01
  ga naar x: 0 y: 0
  richt naar 0 graden
  herhaal
    maak kloon van mijzelf
    neem 10 stappen
    wacht 0.1 sec.
```

8

De lengte van de slang

Gedurende het spel kan de lengte van de slang variëren. Hiervoor moeten we een variabele maken. Dit gaat als volgt:

1. Klik op data
2. Klik op Maak een variabele
3. Vul naam van de variabele in
4. Klik op alleen voor deze sprite
5. Klik op OK



Voeg nu het commando **Data** → **Maak** toe aan de code voor de slang.



```
wanneer vlag wordt aangeklikt
  maak grootte 20 %
  verander uiterlijk naar slang-01
  ga naar x: 0 y: 0
  richt naar 0 graden
  maak Lengte 5
  herhaal
    maak kloon van mijzelf
    neem 10 stappen
    wacht 0.1 sec.
```



9

Kloon code

Nu moeten we ervoor zorgen dat het spoor van kopieën niet langer wordt dan lengte.

Om een kloon code uit te laten voeren moeten we het commando **Besturen** → **wanneer ik als kloon wordt gestart** toevoegen. Dit is de start van het derde code blok voor de slang.

Bouw nu de rest van de kloon code, zoals hiernaast staat afgebeeld.

De code is een weer een eindeloze lus. In de lus staat een wacht, om hem gelijk te laten lopen met de lus voor het bewegen van de slang. Hierna verminderen we de lengte met 1. Is de lengte kleiner dan 1 (0) geworden, dan verwijderen we de kloon weer.



```
wanneer ik als kloon start
herhaal
  wacht 0.1 sec.
  verander Lengte met -1
  als Lengte < 1 dan
    verwijder deze kloon
```

10

Af door rand aan te raken

Het begint al ergens op te lijken. Alleen een spel is pas een spel als je ook af kunt gaan.

Dit doen we door te testen of we de rand aanraken. Het blok dat we daarvoor gebruiken is **waarnemen** → **raak ik (muisaanwijzer) ?**.

Als we de rand aanraken zegt de slang Game Over en daarna wordt ons programma gestopt.



```
wanneer vlag wordt aangeklikt
  maak grootte 20 %
  verander uiterlijk naar slang-01
  ga naar x: 0 y: 0
  richt naar 0 graden
  herhaal
    maak kloon van mijzelf
    neem 10 stappen
    wacht 0.1 sec.
    als raak ik rand ? dan
      zeg Game Over 3 sec.
      stop alle
```



11

De Sinaasappel

In het spel kan de slang sinaasappels eten. Deze sinaasappels verschijnen om de zoveel tijd op een willekeurige plaats op het speelveld.

Als we op de Sprite van de sinaasappel (orange) klikken krijgen we het, nog lege, code veld van de orange te zien.

Het blok **Functies** → **willekeurig getal tussen** geeft een willekeurig getal tussen de twee opgegeven waarden.



```
wanneer vlag wordt aangeklikt
  maak grootte 50 %
  herhaal
    wacht 5 sec.
    ga naar x: willekeurig getal tussen -230 en 230 y: willekeurig getal tussen -170 en 170
    maak kloon van mijzelf
```

12

Eten

Leuk al die op willekeurige plaatsen verschijnende oranges. Maar daar hebben we nog niks aan.

Als de slang een orange raakt moet de orange verdwijnen en de slang wat langer worden. Dat eerste doen we nu. Het langer worden doen we in de volgende stap.



```
wanneer vlag wordt aangeklikt
  verdwijnt
  maak grootte 50 %
  herhaal
    wacht 5 sec.
    ga naar x: willekeurig getal tussen
    maak kloon van mijzelf
```



```
wanneer ik als kloon start
  verschijnt
  herhaal
    als raak ik slang-01 ? dan
      verwijder deze kloon
```




13

Langer worden

Om de slang langer te laten worden moeten we de variabele lengte groter maken. Omdat lengte sprite gebonden is kan dat alleen vanuit de slang code.

We laten de orange dus een signaal uitzenden. Z'n signaal kan door alle andere sprites ontvangen worden. Aan de code voor de slang voegen we dus een blokje toe.



14

Niet kruisen (1)

Zoals het spel nu is kan de slang gewoon door zichzelf heen glijden. Dit is niet de bedoeling. Maar de raak ik sprite? Functie kunnen we niet gebruiken, omdat de slang zichzelf altijd aanraakt.

Maak een variabele (zie stap 8 voor hoe) genaamd Stappen en pas de slang kloon code aan zoals hiernaast.




15

Niet kruisen (2)

Nu het grootste deel van de slang een andere kleur krijgt dan de kop kunnen we testen of de kop de staart kleur raakt.



Om de juiste kleur te kiezen nemen we de stappen zoals in het volgende plaatje weergegeven.



16

Opruimen

De variabelen die wij gemaakt hebben staan iets wat hinderlijk in het speelveld. Als we in de slang code op de data groep kiezen, zien we de variabelen staan met een vinkje ervoor. Klik op het vinkje voor Stappen om hem weg te halen uit het speelveld.

Klik nu met de rechter muisknop in het speelveld de variabele lengte en kies **groot uitlezen**. Verplaats het oranje vlakje door er met links op te klikken en vervolgens te slepen.



17

Alle code tot nu toe

De code voor de slang:

```
when green flag clicked
  make size 20 %
  change appearance to slang-01
  go to x: 0 y: 0
  direction 0 degrees
  make length 5
  loop
    make clone of myself
    take 10 steps
    wait 0.1 sec.
    if raak ik rand? or raak ik kleur? dan
      say Game Over 3 sec.
      stop alle

when I start as a clone
  make steps 0
  loop
    change steps with 1
    if steps > 2 dan
      change appearance to slang-02
    wait 0.1 sec.
    change length with 1
    if length < 1 dan
      delete this clone

when green flag clicked
  loop
    if key pressed left? dan
      direction -90 degrees
    if key pressed right? dan
      direction 90 degrees
    if key pressed up? dan
      direction 0 degrees
    if key pressed down? dan
      direction 180 degrees

when I receive signal HapGenomen
  change length with 1
```

De code voor de sinaasappel:

```
when green flag clicked
  disappear
  make size 50 %
  loop
    wait 5 sec.
    go to x: willekeurig getal tussen -230 en 230 y: willekeurig getal tussen -170 en 170
    make clone of myself

when I start as a clone
  appear
  loop
    if raak ik slang-01? dan
      send signal HapGenomen
      delete this clone
```




Klaar

Je hebt een werkend computer spelletje gemaakt.

Extra

Een spel is nooit echt af. Zelfs als het klaar is. Moet het iets moeilijker? Iets makkelijker. Kan het grafisch iets mooier? Of moet het geluidje toch iets harder of zachter. Er is altijd wel iets.



Bij ons spel zijn er nog een heleboel ruwe randjes. In de volgende stappen worden veel daarvan aangepakt.

De volgende stappen zijn voor diegene die het spel nog iets mooier of beter willen maken. Sommige blokken zijn simpele aanpassingen. Anderen zijn wat uitdagender.

18


Computers en kleuren

De slang is nu twee kleurig. De tweede kleur is duidelijk maar niet mooi. We gaan de slang helemaal zwart maken. Maar we gebruiken de kleur van het lijf om te testen?

Gelukkig voor ons kent de computer heel erg veel kleuren. Echt zwart, en vele tinten die er zo dicht bij liggen dat wij ze ook als zwart zien. Voor de computer zijn het echter verschillende kleuren.

Volg de aanwijzingen op het de afbeelding hiernaast om slang-02b te laden ###pad . Maak de naam van het uiterlijk weer slang-02 (nummertje 4 in de afbeelding

Nu moeten we alleen nog de kleur waarop we testen of de slang zichzelf raakt aanpassen (zie blok 15)



19

Een staart

Laad de afbeeldingen slang-03, slang-04 en slang-05 als 3de, 4de en 5de uiterlijk voor de slang. (Zie blok 1 voor hoe te laden)

Pas vervolgens de slang-kloon code aan zoals hiernaast aangegeven.




20

Achtergrond aanpassen

De achtergrond is nu erg wit en saai. Maar deze kunnen we makkelijk aanpassen. Klik op het aangeven icoontje en kies een van de afbeeldingen uit de Scratch bibliotheek. Voor de uitleg hebben we de achtergrond purple gekozen.

De achtergronden worden na het laden op dezelfde manier weergegeven als de uiterlijken van een sprite. Je kunt ze dus ook op dezelfde manier wissen (zie blok 18)

Test het spel met de nieuwe achtergrond. Is alles nog goed te zien?



21

Snelheid makkelijk afstellen

De snelheid waarmee de slang beweegt wordt bepaald door twee dingen. Het aantal stappen dat de slang neemt en hoelang de wacht is tussen de stappen. De wacht komt voor bij de kop en bij de kloon.

Als we de wacht tijd in een variabele(beschikbaar voor alle sprites) zetten hoeven we bij het aanpassen van de snelheid de wachttijd maar op één plaats aan te passen.

The image shows two Scratch code blocks. The left block is triggered by a green flag click and sets up variables for 'WachtTijd' (0.1), 'StapGrootte' (10), and 'grootte' (20%). It moves the snake to (0,0) and sets its direction to 0 degrees. It then enters a loop that clones itself, moves the clone by 'StapGrootte' steps, waits for 'WachtTijd' seconds, and decreases the length by 1. The right block is triggered when a clone starts and sets 'Stappen' to 0. It enters a loop that changes 'Stappen' by 1, and if it reaches 3, it changes the snake's direction to 'slang-02'. It then waits for 'WachtTijd' seconds, decreases 'Lengte' by 1, and if it reaches 1, it removes the clone. Finally, if 'Lengte' is less than 3 and 'Stappen' is greater than 3, it moves to the next direction.

In het voorbeeld hebben we ook een variabele voor de stap grootte gemaakt. Zo hebben we alle variabelen om de snelheid van het spel te tweaken makkelijk bij elkaar staan.


Dit is vooral handig als je spel groter en ingewikkelder wordt.



22

Sinaasappel verschijnt niet in slang

De sinaasappel verschijnt op een willekeurige plaats. Soms ook midden in de slang. Wat je alleen merkt doordat de slang dan wat langer wordt. Laten we er voor zorgen dat de sinaasappel nooit in de slang verschijnt.




```
wanneer vlag wordt aangeklikt
  verdwijnt
  maak grootte 50 %
  herhaal
    wacht 5 sec.
    ga naar x: willekeurig getal tussen -230 en 230 y: willekeurig getal tussen -170 en 170
    herhaal tot niet raak ik slang-01 ?
      ga naar x: willekeurig getal tussen -230 en 230 y: willekeurig getal tussen -170 en 170
  maak kloon van mijzelf
```

23

Sinaasappel niet eindeloos

Nu blijft de sinaasappel eindeloos staan, totdat ie wordt gegeten. Late we dat wat spannender maken zodat de sinaasappel na verloop van tijd weer verdwijnt.

Dit doen we door bij de sinaasappel een 2^e **Wanneer ik als kloon start** code blok te maken, met een **wacht** en een **verwijder kloon**



```
wanneer ik als kloon start
  wacht 10 sec.
  verwijder deze kloon
```



24

Gevaarlijk voer

Voor dit blok is het noodzakelijk dat je blok 21 ook gedaan hebt.


Klik nu in het sprites veld met de rechtermuis toets op de Orange en selecteer **kopie maken**.

Selecteer de kopie. Klik op de **i**(1), vul een nieuwe naam in(2) en klik op terug(3).

Selecteer nu uiterlijken, van de kopie. Voeg een ander uiterlijk toe. In het voorbeeld hebben we speed.png geladen ###path En verwijder het oude uiterlijk (rechtermuis toets en dan verwijderen).

Ga nu naar de code en pas de wacht tijden aan. Geef ook het signaal dat de kloon verzend als de slang wordt geraakt een andere naam.


Voeg tenslotte het volgende code blokje toe aan de code van de slang.



25

Geluid

Geen spel is af zonder geluid. Dus ook wij gaan wat geluiden toevoegen. Hiervoor gebruiken we het commando start geluid. Bij het commando start geluid kun je de geluiden uit de lijst kiezen. Het toevoegen aan de lijst werkt ongeveer zoals het toevoegen van een afbeelding. Bij (3) kun je het geluid luisteren. Bij (4) wordt het geselecteerd. En bij (5) wordt het geselecteerde geluid aan de lijst toegevoegd.



```
wanneer vlag wordt aangeklikt
  maak WachtTijd 0.1
  maak StapGrootte 10
  maak grootte 20 %
  verander uiterlijk naar slang-01
  ga naar x: 0 y: 0
  richt naar 0 graden
  maak Lengte 5
  herhaal
    maak kloon van mijzelf
    neem StapGrootte stappen
    wacht WachtTijd sec.
    als raak ik rand of raak ik kleur dan
      start geluid chee chee
      zeg Game Over 3 sec.
      stop alle
```

```
wanneer ik als kloon start
  maak Stappen 0
  herhaal
    verander Stappen met 1
    als Stappen = 3 dan
      verander uiterlijk naar slang-02
    wacht WachtTijd sec.
    verander Lengte met -1
    als Lengte < 1 dan
      verwijder deze kloon
    als Lengte < 3 en Stappen > 3 dan
      volgend uiterlijk
```

```
wanneer vlag wordt aangeklikt
  herhaal
    als toets pijltje links ingedrukt dan
      richt naar -90 graden
    als toets pijltje rechts ingedrukt dan
      richt naar 90 graden
    als toets pijltje omhoog ingedrukt dan
      richt naar 0 graden
    als toets pijltje omlaag ingedrukt dan
      richt naar 180 graden
```

```
wanneer ik signaal HapGenomen ontvang
  verander Lengte met 1
  start geluid pop
```


```
wanneer ik signaal Versnel ontvang
  maak WachtTijd 0
  start geluid bird
  wacht 8 sec.
  maak WachtTijd 0.1
  start geluid chord en wacht
```




26

Code Overzicht

De complete code voor de slang staat op de vorige bladzijde.




```
wanneer vlag wordt aangeklikt
  verdwijnt
  maak grootte 50 %
  herhaal
    wacht 5 sec.
    ga naar x: willekeurig getal tussen -230 en 230 y: willekeurig getal tussen -170 en 170
    herhaal tot niet raak ik slang-01 ?
      ga naar x: willekeurig getal tussen -230 en 230 y: willekeurig getal tussen -170 en 170
  maak kloon van mijzelf
```



```
wanneer ik als kloon start
  verschijnt
  herhaal
    als raak ik slang-01 ? dan
      zend signaal HapGenomen
      verwijder deze kloon
```



```
wanneer ik als kloon start
  wacht 10 sec.
  verwijder deze kloon
```



```
wanneer vlag wordt aangeklikt
  verdwijnt
  maak grootte 50 %
  herhaal
    wacht 9 sec.
    ga naar x: willekeurig getal tussen -230 en 230 y: willekeurig getal tussen -170 en 170
    herhaal tot niet raak ik slang-01 ?
      ga naar x: willekeurig getal tussen -230 en 230 y: willekeurig getal tussen -170 en 170
  maak kloon van mijzelf
```



```
wanneer ik als kloon start
  verschijnt
  herhaal
    als raak ik slang-01 ? dan
      zend signaal Versnel
      verwijder deze kloon
```



```
wanneer ik als kloon start
  wacht 9 sec.
  verwijder deze kloon
```

